Adam News Network

EOS Programmer's Manual

Here it is! The EOS Programmer's Manual by Guy Cousineau. It's a very good document, and |
hope that you find it useful.

This document has been reprinted with the permission of the ADAM News Network. This
document is for personal use only. Please note that a lot of time and effort has been put into this
manual by Guy Cousineau (the original author), and by myself (OCR, formatting, and error
checking). If you intend to use this manual for commercial use, please ask permission first!

FOREWORD

The EOS (Elementary Operating System), is one of three operating systems available to the ADAM programmer.
The other two, (CP/M (now replaced by the superior and user friendly T-DOS), and 057), are best suited for "work
horse" or practical programs, and game graphics, respectively.

The EQOS, the subject of this work, is a sort of bridge between the extremes of the "work horse” T-DOS, and the
graphics game 0S7 operating systems. Thus the three operating systems complement one another.

Inasmuch as it is in the best interest of ADAM owners to have as many good programs for the ADAM as possible,
(thus extending its usefulness, and thereby its practical lifetime), the ADAM News Network, (ANN), felt that it was
desirable to publish a work that would help all interested parties write programs with the greatest facility that could
be made available to them. ANN wants to encourage the ADAM owner to explore the field of Machine Language
(ML) programming, and to encourage the development of ML programs. This work is designed to help ADAM
owners use the EOS routines as correctly, and as effectively as possible.

Thus, through the generous work of Guy Cousineau, the author, this work is presented to help all who would, begin
programming efforts to the benefit of us all.

The EOS, of course, must be addressed via ML instructions. There is, on the other hand, much to be said of the
advantages of programming in the BASIC language. It is certainly easier to understand initially. But the direct use of
ML programming instructions, though more time consuming perhaps, allows the writer to have program
compactness, flexibility, and speed of execution, not possible by the use of any other programming language.
Furthermore, some of the hardware specific functions on the ADAM can only be addressed by ML routines.

ANN also wishes to convey to all readers. and interested parties the understanding that the publication of this work
is not the culmination of an EGO TRIP on the part of ANN or on the part of the author. If there are things found
herein that are unclear, confuse, or clutter the mind, the reader is encouraged to write to, or otherwise get in touch
with the ADAM community. Call the FIDO NET ADAM conference, any ADAM BBS, Compuserve, and direct
whatever question you may have to the author or to any other ADAMite who might be in a position to answer such
queries or implement your suggestions. No one will be offended. In fact, the contrary is the case:

ANN and its members encourage such dialogue.



Whether or not other works outlining the best uses of the 0S7 and the T-DOS operating system will. be published in
the future will depend upon the response to this publication. ANN is dedicated to the promotion of anything that will
help ADAM users. If a need is shown for such works, they will surely be forthcoming.

Having been requested to write this "INTRODUCTION", | would like to say that we are indeed fortunate to have
Guy Cousineau as the author. | know Guy, and have worked with him on projects; | have always found him to be
very personable and understanding. | personally have had to go back for repeated explanations of some items which
he has written in the past, and found that he is not an AUTHORITY but a TEACHER. He has patience beyond
reason, and tries very hard to help whenever he can. He is indeed an expert, but he is not an AUTHORITY who
cannot bear to have his writings and. ideas questioned.

In his work, Guy does an excellent job of making the EOS routines understood. | have an original EOS manual, and
have noticed that there are many invalid assumptions made there relative to the level of understanding of the reader.

Guy has clarified it greatly where things were nebulous.
We are fortunate indeed!
Mel Ostler

[ADDRESS REMOVED]

OVERVIEW

What is the EOS? It is the Elementary Operating System of the ADAM, and not the Extended Operating System, as
some people call it. It consists of a series of routines which help the programmer interface with the hardware

attached to the ADAM. The EOS takes up the top 8K of memory and uses an additional 3K of buffers just below it.
This 8K section has routines to address the keyboard, printer, screen, and storage devices. Each routine has a
specific calling convention and may return information about the status of the operation or of the operating system
itself.

The EOS is used by applications programs such as SmartBasic, AdamCalc, SmartFiler, etc. Other programs such as
FileManager and other AJM software products (this is the only plug you will see in this manual) make use of the
EOS to handle its elementary functions.

This manual talks a bit about the benefits of machine language programming, the EOS structure, and memory
configuration. The greater part of the manual, however is devoted to the explanation of EOS routines. Not all
routines will be covered. Some of the incomplete or non-implemented routines will be skipped so not to confuse the
reader. It is intended as a reference tool for the machine language programmer, whether beginner or advanced. The
first section deals with executive calls. As most application programs do not require these, it is recommended to skip
over this section and come back to it later. Otherwise, you might try to absorb a lot of technical information which is
not required. It has been placed first since it is logical in the sequence of operation.

Considering that Assemblers usually generate HEXADECIMAL codes, all addresses quoted in this manual will be
in hexadecimal format only. This manual does not propose to explain Z-80 programming or the use of the Op-
Codes. Someone with a knowledge of Z-80 programming will be able to use the manual to effectively access the
operating system. In addition, some of the examples and complete programs/routines supplied in this manual may
help the programmer pick up tricks about Z-80 programming or the EOS itself.

Please address any technical questions to:
Guy Cousineau

[ADDRESS REMOVED]



MACHINE LANGUAGE PROGRAMMING
Why use machine language programming? My three favourite reasons are SPEED, SIZE, and CONTROL.

By skipping superfluous steps and compressing operations into the most effective structure, you can achieve
processing speed which has no match in any language, whether it be BASIC, Fortran, C, Pascal, etc. Think of the
fastest BASIC routine you have seen to sort a series of strings. Then look at the speed of ADAMCALC's sort
feature. The difference is amazing.

The programmer can effectively manipulate routines and common subroutines into a package that occupies less
space than any other language. While some powerful BASIC programs may be quite short, consider that they have
27K of overhead: the BASIC itself! A Self-contained Machine language program can be written using only the EOS
to handle console and file I/0. | have seen some self-contained machine language programs which take less than 1K
and perform several functions.

Control is another key aspect of machine language programming. There is no need to assume that a particular
SYSTEM routine (like BASIC's FOR, INT, etc.) will perform as they should. The programmer writes all his own
support routines which perform the operation in the manner that suits the programmer's purpose. It is a bit more
demanding but the outcome is usually a very efficient and compact program.

Bear in mind that when working with any language, the computer will ALWAYS do what you tell it to do; this is
not always the same as what you intend it to do. It is much easier to throw the computer into a death spin when
writing in machine language: you don't have SmartBasic checking your syntax and reporting other errors.

EOS STRUCTURE

The EOS uses a jJump table to gain access to its functions. A jump table is a series of standard entry points which
then pass control over to the routine that actually does the work The advantage of a jump table is that revisions to
the operating system do not affect the addresses that the programmer uses to access the system's functions. The EOS
routines can be subdivided into 9 categories, Each is composed of several routines to perform a specific series of
tasks:

Executive Calls

These are the high level operations such as startup, device scanning, initialization, etc. Most of these
routines are used on a cold boot to get the computer in running order. Additionally, self booting software
will make use of some of these routines to set up the computer according to the programmer's requirements.

Console (Screen) Output

These routines take care of printing characters on the screen. They make use of the Video Ram routines
outlined below.

Printer Interface

These routines take care of printing characters on the ADAM printer. Although parallel printer interfaces
are available, the EOS had made no provision for this feature.



Keyboard Interface

These routines take care of fetching characters from the keyboard.

File Operations

These routines take care of disk/tape input output at the file level. Files can be created, opened, read,
written to, closed, deleted, etc. in a fashion equivalent to the same commands in SmartBasic.

Device Operations

These are the routines which interact directly with the devices attached to the ADAM. They include the
keyboard, printer, disk drives and tape drives. The Video Ram is not considered a device by the EOS.

Video Ram Management

These routines handle the movement of information (characters, sprites, shapes, colour, graphics, etc.) to
and from video memory. It is these routines which actually put something on the screen.

Game Controllers

Since the joysticks are not devices, these routines handle the reading and decoding of information from the
joysticks.

Sound Routines

The EOS has complex routines which are used by some software (like the arcade games) to generate
elaborate sound effects.



MEMORY USAGE

FROM

D390

D400
EQ0O
E1F6
E2C7
E618
F3DA
F446
F4BA
F4FC
F5DC
F832
FASE
FBFF
FC30
FD60
FECO
FEC4

TO

D3FF

DFFF3
E1F5
E2C6
E617
F3D9
F445
F4B9
F4FB
F5D4
F831
FASD
FBFD
FC2F
FDSF
FEBF
FEC3
FFFE

CONTENT

File Control Blocks

1K blocks used for directory and file I/O
Video Ram routines

Joystick routines

Sound routines

File routines

Data used to INIT a medium

Device routines

Keyboard routines

Printer routines

Screen routines

Executive routines

Device Control Block Routines

Data tables

JUMP TABLES

Various data tables, storage, and stack
Processor Control Block

Device Control Blocks (one for each device)

The average program can use all memory below D390. This leaves the full EOS available to perform any function
which many be required.

A program which requires additional memory may choose to use its own buffers for read/write operations. If it does
not make use of the FILE routines, it may overwrite up to DFFF and gain an additional 3K of usable memory.

When even more memory is required, An application program may supply its own video drivers, and joystick
routines (if required). It can then use memory up to F445 and gain another 5K. This leaves just the raw device
drivers in the EOS. Provided no complicated video routines are required, this can be a viable alternative for very
large programs.



EOS ROUTINES

This is the start of the technical section of the manual. It describes routines which can be accessed form the EOS
jump table. For the sake of uniformity and ease of reading, each routine description will be on one page with the title
of the routine at the top of the page. Descriptions will use the following format:

JUMP TABLE ADDRESS: The address to CALL. If you want to determine the
actual address of the routine, skip the first byte at this address (the JUMP
instruction) and extract the REAL address from the next two bytes.

ENTRY: A Description of values which must be
BC placed in registers to tell the function
DE what to do. Only the relevant registers
HL will be shown.
EXIT: A Some routines return information. The exit
BC values will help you make the best use of the
DE information provided. You will also note that
HL some routines PRESERVE register values. This can

be used to your advantage to save program space.

DESCRIPTION:

The description section will usually be in two parts: the first describes the purpose of the routine the second explain
how the routine does its job. When a routine calls another major EOS routine, its name will usually be shown in
UPPERCASE along with the page number in brackets.

EXAMPLES:

Where it can help illustrate the use of a routine, examples will be provided. These will be in machine language
mnemonics with comments in the right margin.

10



EXECUTIVE CALLS

INITIALIZE EOS
JUMP TABLE ADDRESS: FC30
ENTRY: none
EXIT: B boot device number

DESCRIPTION: = = =

When you turn on the computer or pull the reset switch, the EOS is entered at this routine. This is the one that gets
the computer in working order by setting the stack, initializing tables, setting up the devices, and checking for the
presence of a boot tape or disk. While this function is used only at a cold start, a programmer may use it to reboot by
prompting for an "insert media” and calling FC30 to clear memory and boot the media.

The routine starts by setting the EOS stack. It then nulls out all the data tables starting at FD6I. The EQS revision
number is written to FD60O. It calls SET VDP PORTS(83) and SOUND OFF(104). It then nulls all the video RAM
with a call to FILL VRAM(95) and switches in the normal memory configuration with a call to BANK
SWITCH(20). The next step is to perform a HARD INIT(12) to set all the devices. Next is a call to INITIALIZE
FILE MANAGER(41) which sets up the file management buffers. It then scans for the presence of media in disk 1,
disk 2, tape 1, and tape 2 in that order. The first to contain media is presumed to be the boot media. Block 0 of that
media is loaded at C800. The next step is very important; The device number of the boot media is placed in register
B and a jump is made' to G800 where the media boot code can be executed. Boot code should store the boot device
number so further media activity can be performed on the default media. If no boot media is found, a jump is made
to the memory resident processor (electronic typewriter) via the JUMP TO SMARTWRITER(19) routine.

EXAMPLES:

To prompt the user to insert a disk or tape and perform a boot, you can use the following instructions:

LD HL, INSERTMSG ;prompt message.

CALL PRINT ;your print to screen routine.

GALL FCo6G ;read keyboard (wait for character).
JP FC30 ;do the EOS boot.

If you wish to boot a disk or tape without resetting the EOS tables or performing an initialization, you can replace
the last instruction with a jump to F86A. Note that this is the direct address for EOS revision 5 and that it may not
work on other revisions to the EOS. The BOOT device is still returned in B to the Boot block of the media. It is up
to the programmer to pass the boot media to the main program. You could store it at (FD6F) which is the EOS-5
address of current device.

11



EXECUTIVE CALLS

HARD INIT

JUMP TABLE ADDRESS: FC5D

ENTRY: none

EXIT: A destroyed
BC \
DE preserved
HL /
DESCRIPTION:

The function of this routine is to set the processor control block which in turn controls the allocation of device
control blocks. It synchronizes the Z-80 processor with the 6801processor which controls the AdamNet.

This routine starts by initializing the Processor Control Block to FECO. Compare to the SOFT
INITIALIZATION(18) routine which allows the user to set the PCB. It calls HARD RESET NET(13) and the
DELAY (14) to initialize the system. The device control blocks are nulled out and a call to SYNCHRONIZE
CLOCKS(I5) is made. The final step is a call to SCAN FOR DEVICES(16) to allocate a DCB for each found device.

Although it would be considered a drastic measure, this routine could be called to effectively reset the ADAM and
re scan all the devices.

EXAMPLES: none

12



EXECUTIVE CALLS

HARD RESET NET
JUMP TABLE ADDRESS: FCo60
ENTRY: none

EXIT: A Zero
C reset port number
others unchanged

DESCRIPTION:

This function does a hardware reset on the AdamNet. It gets the Net port number from location FC28 and sends a
reset request (OF). It waits awhile and sends an idle command to the net. The routine returns the port number in
register C but it is very unlikely that you will ever need this information.

This routine can be called to reset the net without clearing all the DCB's

EXAMPLES: none

13



JUMP TABLE ADDRESS:

ENTRY: none

EXIT: A

DESCRIPTION:

others

EXECUTIVE CALLS

DELAY AFTER HARD RESET

FC3C

0

preserved

This routine pauses for 114 clock cycles. It is used after a network reset to make sure that the devices have time to
idle down. With 3 million clock cycles per second, 114 is a blink of an eye for us but almost an eternity for the Z-80
processor. For some reason, the EOS designers did not include this delay as part of the HARD RESET NET(13)

routine and the delay MUST be called for proper timing.

While the routine as-is has limited value, you can make use of it to create your own timed delays by patching
directly into the routine. Remember the default values so you can restore them when you are through. Firstly,

address F962 has a default value of 1. For the delay to have any substance, this should be reset to 0. Address F965
has a default value of 0. This is the one we can modify to create delays ranging up to 2.5 minutes.

EXAMPLES:

LD
LD
LD
LD
CALL
LD

instruction) .

INC
LD

VALUE
2

4

7

9

18

36

DELAY

A0
(F962)

A,53
(F965)
FC3C
(F965)

(F962)

VALUE
53

71

88
106
212

DELAY
30

40

50

60
120

14

1/2 minute

1 minute
2 minutes

;set 30 second delay.
;do it.
;note that A is zero (save

an



EXECUTIVE CALLS

SYNCHRONIZE CLOCKS

JUMP TABLE ADDRESS: FCB1
ENTRY: none

EXIT: zero flag indicates success
A error code if NON-ZERO
others preserved

DESCRIPTION:

The purpose of this routine is to get ADAM's internals synchronized. Two synchronize requests are sent through the
Processor Control Block. It is essential that these requests function properly. Otherwise, the ADAM NET may not
function properly. The Synchronize Clock request must be sent each time the network is reset.

One of the byproducts of this function is to cancel out All active device control blocks. Thus a SCAN ACTIVE must
be done after each synchronize clock function call.

EXAMPLES:
CALL FCB1 ;request the synchronize.
JR Z,CO0D ;the synchronization went ok.

;here you must decide how to handle the error
;you can call the function again or abort entirely

telling the user that ADAM has serious problems

GOOD:

CALL FC8A ;scan for active devices.

15



EXECUTIVE CALLS

SCAN FOR DEVICES
JUMP TABLE ADDRESS: FC8Aa

ENTRY: the Processor Control Block must be set

EXIT: A Zero
others preserved

DESCRIPTION:

The purpose of this function is to find all the active devices on ADAM NET. It begins by zeroing out all Device
Control Blocks. This will effectively remove devices which are no longer on line. On the other hand, the scan will
pick up devices which have been powered up after the ADAM has turned on. Thus you should make sure that all
devices are turned on before calling this function.

An area of 314 bytes is cleared above the 4-byte processor control block. It represents the 15 DCB's which are 21
bytes each. A count is kept in PCB+3 of the total number of devices found and a 21-byte DCB is set up for each
device found. If you wish to check the number of devices found, you will have to FIND the PCB and examine its

byte number 3.

EXAMPLES:
CALL FC8A ;set up the DCB's.
CALL FC5A ;get the PCB address into IY.
LD A, (IY+3) ;number of devices.

;now you can tell your program or viewer

;how many devices you found

16



EXECUTIVE CALLS

RELOCATE PCB

JUMP TABLE ADDRESS: FC7B
ENTRY: HL Address to relocate PCB to
EXIT: A 83H zero flag set

BC preserved

DE preserved

HL new PCB address
DESCRIPTION

This function is used to relocate the Processor Control Block. Although | can see no particular reason for this
function, COLECO must have thought that it might be a requirement with some of the expansion hardware or for
special purpose applications.

The routine begins by getting the current PCB location since commands must still be sent through the current PCB
location. The new address is written to the PCB data area and a SET command is sent to the NET. The routine then
loops endlessly waiting for the net to acknowledge. If it does not, the system will hang up; there is no time out for
this operation. The last step is updating the CURRENT PCB location in memory.

Once the PCB has been relocated, all devices are effectively put off line. You have to issue a SCAN FOR DEVICES
(16) to restore all devices. | am not sure if it is necessary to perform a hard reset, but to be on the safe side, | would
recommend using the SOFT INITIALIZATION(18) instead of this routine to relocate the PCB.

EXAMPLES:
LO HL, 8000H ;address to relocate to.
CALL FCT7B ;move the PCB please.
CALL FC8A ;scan for devices.

17



EXECUTIVE CALLS

SOFT INITIALIZATION

JUMP TABLE ADDRESS: FC8D
ENTRY: HL New PCB address
EXIT: A destroyed

others wunchanged

DESCRIPTION:

The function of this routine is to set the processor control block which in turn controls the allocation of device
control blocks. It synchronizes the Z-80 processor with the 6801 processor which controls the AdamNet.

This routine starts by initializing the Processor Control Block to the value supplied in HL. After that, its execution is
virtually identical to HARD INIT(12). It does a hard reset, a delay, a synchronize, and a scan for devices.

Since you need to do all these housekeeping functions when you want to relocate the PCB, it is best to call this
vector instead of RELOCATE PCB.

EXAMPLES:

18



EXECUTIVE CALLS
EXIT TO SMARTWRITER
JUMP TABLE ADDRESS:FCE7

ENTRY: none

EXIT: none

DESCRIPTION:

This is the routine used by the COLD BOOT sequence (power up) when no bootable media is found. It bank
switches in the SMARTWRITER ROM and jumps to it. If you wish to abort a program you can call this routine to
effectively halt all operations.

Since this routine does a BANK switch, it is essential that the STACK be located in upper memory when it is
invoked. Although it is not a standard location, | suggest using 65535 since it is used only temporarily and the
SmartWriter will set its own stack when it takes over.

EXAMPLES:
CALL ABORTYN ;ask user to abort.
RET NZ ;changed his mind.
LD SP,FFFF ;set stack to upper half.
Jp FCE7 ;let's get out.

EXECUTIVE CALLS

SWITCH MEMORY BANKS

JUMP TABLE ADDRESS: FD14
ENTRY : A desired memory configuration
EXIT: A current configuration

B current configuration

C memory port number

DE preserved

HL preserved

DESCRIPTION:

This routine can switch either the upper or the lower 32K of memory to any of 4 configurations. It simply sends the
requested configuration to the memory bank switch port. Although it is a simple routine, it is easy to get into trouble
with it. When the memory configuration is changed, program control may wind up in the switched bank with
unpredictable results. Since the EOS is in the top half of memory, the TOP HALF should never be switched using
this routine. If you do, the EOS will effectively disappear, and your program will crash. It is also essential that the
Routine which calls this function be also located in the upper half of memory (above 8000H) or program control will
also be lost.

Consult the appendix for ADAM's memory bank configuration.
EXAMPLES:

This is the breakdown of the routine used to switch banks. If you wish to switch in the upper half of memory, you
will require a routine similar to this in the lower 32K to make the switch.

LD A, (FC27) ;get the port number.

LD (SAVE) ,a ;save the address in lower memory
somewhere.

LD C,A ;put port in C.

LD A,config ;put in the configuration you want.

ouT (C) ;switch it over.

;do what you wish to do in here

LD A, (SAVE) ;get the port back.

LD C,A

LD A,normal ;bring the EOS back into top half.
ouT (C) ;send request to port.

CONSOLE OUTPUT

INITIALIZE CONSOLE

JUMP TABLE ADDRESS: FC36
ENTRY : B number of columns (0 to 31)
C number of lines (0 to 23)
D home column
E home row

HL pointer to pattern name table

EXIT: all registers lost
DESCRIPTION:

This routine is used to set up a WINDOW for screen display. Registers B and C contain the, number of rows and
columns while registers D and E contain the upper left corner of the window. The other parameter required is base
address of the Pattern Name Table. If you have previously set up VDP, you should have NOTED what that address
was. If you; are using a routine in conjunction with SmartBasic, the default pattern name table address is 1800H.

The routine stores lines and columns, sets up minimum and maximum values for ROW and COLUMN based on the
supplied parameters. You can set up many windows and jump around between them by repeating calls to this
routine. When this routine exits, the default cursor (an underline) is placed in the top left corner of the window. You
can then send a move cursor command (see console display page 18) to place it at the appropriate location in the
window.

EXAMPLES:

This routine sets up a 12 line window in the centre of the screen:

LD B, 20 ;20 columns.

LD Cc,12 ;12 rows.

LD D, 6 ;home column.

LD E, 6 ;home row.

LD HL, 1800H ;or whatever your pattern address is.
CALL FC36 ;set up the screen.

CONSOLE OUTPUT.

CONSOLE DISPLAY REGULAR

JUMP TABLE ADDRESS: FC33

ENTRY: A character to send
EXIT: all registers preserved including A
DESCRIPTION:

This routine prints whatever character is in the accumulator to the screen. It presumes that the VDP has been set up
and that a window has been defined. It will print ALL characters including the graphic representation of the control
codes (0-31). If you wish to send a control CODE, use the routine on the next page.

The routine first sends the character to video RAM. Then it advances the cursor position, going to the next line if
required. If the cursor is on the last line, the screen is scrolled.

EXAMPLES:

This subroutine is used to print an incoming message in register HL. It presumes that the string to print is followed
by a null (ASCII 00).

PRTSTR:
LD A, (HL)
OR A
RET Z ;the string was over.
CALL FC33 ;print the thing.
INC HL ;remember HL was preserved.
JR PRTSTR ;loop until end of string.

CONSOLE OUTPUT

CONSOLE DISPLAY SPECIAL

JUMP TABLE ADDRESS: FC39
ENTRY: A character to print or PLACE CURSOR request
D column to go to if PLACE CURSOR
E row ,to go t9 if PLACE CURSOR
EXIT: all registers preserved including A
DESCRIPTION:

This routine, like console display on previous page will print a character on the defined window. It begins however
by checking for 12 special control codes. If it finds one of these, it executes the following control functions:

CONTROL KEYBOARD FUNCTION
CRARACTER EQUIVALENT PERFORMED
08 BACKSPACE move cursor left one
0A &, move cursor down one line (line feed)
0oC AL clear screen and home cursor
0D RETURN return cursor to start of line
(must send line feed if new line wanted)
16 NV delete to end of line
18 "X delete to end of screen
1C N place cursor at position DE
80 HOME home the cursor (no clear)
A0 up arrow move up
Al right arrow move right
A2 down arrow move down
A3 left arrow move left

Note that there is no CURSOR ON or CURSOR OFF. In EOS-5, you can replace 3 bytes starting at F658 with.
ZEROS to turn the cursor off. Be sure to remember the values to turn it back on again.

EXAMPLES:

When the string STRING is printed, it will send the cursor home, skip 2 lines, print HELLO, and clear the rest of the
screen. Note that the first two lines would not be erased by this operation:

LD HL,STRING

REPEAT:

LD A, (HL)

OR A

JR Z,CONT
CALL FC39
INC HL
JR REPEAT
STRING: DB 80H,A2H,A2H, 'HELLO' ,18H,0

CONT:

;Program continues here

PRINTER INTERFACE

PRINT CHARACTER

FC66

JUMP TABLE ADDRESS:
ENTRY: A character to print
EXIT: zero flag set - successful

A error code if NZ

others preserved
DESCRIPTION:

This routine sends one character to the printer. Compare to print buffer on next page.

If the printer is not found, error code number 1 is returned in the accumulator. This means the NET does not know
the printer is there. You should re scan for active devices and if the printer is still not found, you should abort with a
warning to the user.

If the printer is busy, the routine returns error code number 2 in the accumulator. You can wait a bit and retry.

If the printer is off-line (i.e. not working), error code 3 is returned. If this happens, do NOT retry; it will serve no

purpose.

If the printer is ready, the character is placed in the print buffer and the printer is asked to print it via the PRINT

BUFFER routine on the next page.

EXAMPLES:

RETRY:

LD
CALL
JR
AND
DEC
JP
DEC
JR
DEC
JP

GOOD:

;continue program

A, character
FC66
Z,GOOD
127

A,
Z,NODCB
A

Z, RETRY
A
ZPTRDIED

;what you want to print.
;ask printer to do it.

;strip acknowledge bit if any.
;was A one?
;try and find the printer
:was A two?
;let's loop until it works.
:was A 3?
;printer died? Now what do we do?

PRINTER INTERFACE

PRINT BUFFER

JUMP TABLE ADDRESS: FC63
ENTRY: HL pointer to string (terminated with ASCII 03)

EXIT: Zero Flag set - successful
A error code if non-zero
others preserved

DESCRIPTION:

This routine will print a string pointed to by register HL. The bytes are sent to the printer in groups of 16. If the
string (or remainder) is less than 16 then the partial string is sent and then the routine exits. You can use this routine
to print strings of any length: an entire document if you wish. The nice thing about it is that you only need to place a
"03" at the end of the string.

The routine stays in control until the entire string is printed. Since the printer is a very slow device, you may choose
the alternate method using START WRITE and END WRITE explained on the following pages. The error codes are
the same as for PRINT CHARACTER:

1 No printer

2 Printer busy

3 Printer off line (idle)
EXAMPLES:

This is an example of a subroutine used to print a string. The subroutine expects HL to point to the data and takes
care of errors:

LD HL,STRING ;print this.
CALL PRINTIT ;ask subroutine to do it
PRINTIT:
CALL FC63 ;print the whole thing.
RET z ;OK..
AND 127 ;strip acknowledge bit.
DEC A
JR Z,NOPRINT ;No DCB found.
DEC A
JR Z, PRINTIT ;retry.if busy.

;if we get here then the printer is off line.
NOPRINT:

;if we get here then try FINDING the printer again.

PRINTER INTERFACE

PRINTER STATUS

JUMP TABLE ADDRESS: FC84

ENTRY: none

EXIT: Zero flag set - READY
A error code if non zero

IY  address of DCB (only if no error) others preserved

DESCRIPTION:

Although it might be considered a bit of a waste, this is a handy routine to use. You don't need to remember the
DEVICE number of the printer(2) and call REQUEST STATUS.

If you wish to check the availability of the printer prior to sending a character or string, use this routine. When used
in conjunction with the routines on the next pages, it allows you to do other things while you are waiting for the
printer to be READY.

The most common errors are

1 no DCB
2 busy (it is printing something)
3 off line (idle)

EXAMPLES:

This routine waits for the printer to be ready and then sends the character in A to be printed.

LD C.A ;save the character.
RETRY:
CALL FC84 :check status.
JR Z,READY
AND 127 ;strip acknowledge bit if any.
CP 2
JR Z ,RETRY

;process other errors here.

28



READY:

LD
CALL
RET

AC
FC66

;print it.

PRINTER INTERFACE

START PRINT CHARACTER

JUMP TABLE ADDRESS: FCOF

ENTRY: A character to print

EXIT: Zero flag set - successful
A error code if non-zero

others preserved

DESCRIPTION:

This routine is used to set up background printing. If the printer is READY, it asks it to print a character and returns
the control to the user. Since the printer is very slow, you can perform a few other tasks and use END PRINT
CHARACTER to check if the printer is done.

The error codes are the same as the other print routines.

EXAMPLES:

This subroutine expects a character to print in A. It sends it to the printer first (since it is slow); then to the screen.
Finally it waits until the printer is done prior to returning.

LD C,A ;save the character.
CALL FCOF ;start print.
JR NZ, PERR
LD AC ;get character back.
CALL FC66 ;print to screen.
JR NZ,SERR
RETRY:
CALL FC42 ;is printer done?
JR NZ,PERR ;must check error FIRST.
JR NC,RETRY ;if no error and not completed then wait.
RET ;yes we are finished.

PERR: ;process printer error here.

SERR: ;process screen error here.

PRINTER INTERFACE

END PRINT CHARACTER

JUMP TABLE ADDRESS: FC42

ENTRY: none

EXIT: CARRY SET completed
NO CARRY and ZERO not completed
NON ZERO error
A error code if no carry and non-zero

others preserved

DESCRIPTION:

This is the companion routine to start print character. It finds the DCB and returns an error if not found. It will also
return an error if the printer is off line or is busy. If the printer is done, the carry flag is set prior to returning to
caller.

This routine does not retain control while a character is being printed. It allows the programmer to do some co-
processing.

EXAMPLES:

The routine shown in start print buffer is not the most effective way of maximizing throughput since the print
routine does only one task and spends the rest of the time waiting. Once it is done, the programmer still uses
valuable processing time to FETCH another character to send to the routine. Another approach is to send the FIRST
character to print using START PRINT and not check for completion. All subsequent characters can be sent to this
routine which waits for the first to be done prior to sending another.

LD A,CHAR
CALL FCOF ;this is the first character printed.
LD A,CHAR

CALL PRINTC ;all subsequent chars to this routine.

PRINTC:

WAIT:

LD

CALL
JR

JR

LD
CALL
RET

PERR:

C.A :save new character to send.

FC42 ;end print the last one?
NZ,PERR ;o0ps.

NC ,WAIT

AC

FCOF ;start to print this one.
z ;good start.

;here we have a printer problem.

PRINTER INTERFACE

START PRINT BUFFER

JUMP TABLE ADDRESS: FC9C

ENTRY: HL points to a string terminated with ASCII 03
EXIT: Zero Flag set - success
A error code if non-zero

others preserved

DESCRIPTION:

This routine was intended to print up to 16 characters on the printer in background. It requires the use of the
companion END PRINT BUFFER routine on the next page. The routine starts by finding the printer and returns an
error if the printer is not found or busy. It then looks through the input string for a 03 in the first 16 characters. If
none is found, it STARTS to print the first 16 characters. The routine may have been intended to handle longer
strings since it keeps track of where it is (when longer than 16).but does not seem to have been fully implemented.

Since the ADAM printer is slow, you will have about a second and a half (almost an eternity for the Z-80) to
perform some other work while the printer is busy doing this task. You can periodically check if the printer is done
and then send more characters When you send your last 16 (or less), you can just carry on doing something else,
provided you check if the printer is done prior to starting another print sequence.

EXAMPLES:

See example on next page which sends a LONG string to the printer.

JUMP TABLE ADDRESS: FC3F
ENTRY: none
EXIT: CARRY SET
NO CARRY
NON ZERO
A
others
DESCRIPTION:

PRINTER INTERFACE

END PRINT BUFFER

completed
and ZERO not completed
error
error code if no carry and non-zero
preserved

This is the companion routine to START PRINT BUFFER(33). It allows you to perform other tasks
and check if the printer is done the 16 (or less) characters you sent. Once completed (or error), you can send

another with START PRINT BUFFER(33).

EXAMPLES:

The following routine sends a LONG string (which has a 03 at the end) to the printer. It makes use of the wait time
to call another routine which could perform any small task.

PUSH
ID
ID

LD

OR

SRL

RR

SRL

RR

SRL

HL

A3
BC,65535
CPIR
HL,0

A

SBC HL,BC

;save start of string.

:find the 03.

:this is the number of characters

RR L

SRL H
RR L ;this is the number of 16-byte packets to send.
LD B.H
LD CL
INC BC ;correct count
POP HL ;0et back original pointer;
MORE:
CALL FCoC ;start to print.
JR NZ,ERROR
WAIT:
CALL ABCD ;routine better save BC and HL.
CALL FC3F ;is printer done?
JR NZ, ERROR
JR NC,WAIT
LD DE,16
ADD HL,DE ;advance pointer
DEC BC
LD AB
OR C ;is BC zero?
JR NZ,MORE

KEYBOARD INTERFACE

KEYBOARD STATUS
JUMP TABLE ADDRESS: FC81
ENTRY: none
EXIT: Zero flag set - no errors
A error code if non-zero
Y address of DCB (only if no error)

others preserved

DESCRIPTION:

This routine is used to find out if the keyboard is active. It will return a 1 in the accumulator if there is no DCB. This
means that the keyboard was not found. It may indicate hardware failure or just that the keyboard was not plugged in
when the system was turned on. Error code 3 means that the device is idle.

While it may be useful to check the keyboard status in this manner, you still need another EOS call to actually fetch
a character from the keyboard. This routine could be used in a power up sequence to ensure the keyboard is there.

EXAMPLES:

36



KEYBOARD INTERFACE

READ KEYBOARD

JUMP TABLE ADDRESS: FC6C

ENTRY: none

EXIT: Zero flag set - no error
A character if non zero
else it contains the error code
others preserved

DESCRIPTION:

This routine starts by calling START READ KEYBOARD(28). If this fails, an error code is returned. The routine
then keeps calling END READ KEYBOARD(29) until a character or an error is received. The routine remains in
control until a key is pressed; this is like SmartBasic's GET command.

For simple keyboard input, this routine is adequate. It should not be intermixed with the START and END read
commands as you may wind up missing a character.

EXAMPLES:

This routine requests characters from the keyboard and places them in a buffer at (HL) until <KRETURN> is pressed.

LD HL,BUFFER ;point to a memory area for data.
CALL GETENTRY fill it up please

:more program here.....

GETENTRY:
CALL FC6C ;read keyboard.
JR NZ, ERROR ;something went wrong.
LD (HL) ,A ;save character even if <CR>
cp 13
RET z ;yes we have a full entry.
INC HL
JR GETENTRY ;get another character.

KEYBOARD INTERFACE
START READ KEYBOARD
JUMP TABLE ADDRESS: FCA8

ENTRY: none

EXIT: Zero flag set - successful
A error code if non-zero
others preserved

DESCRIPTION:

This routine asks the keyboard to fetch a character. The Keyboard will not acknowledge until it a key has been
pressed. After that, the only thing the programmer needs to do is call END READ KEYBOARD(39) to see if there

is a character waiting.

If there is no DCB or the keyboard is off line, an error code will be returned.

EXAMPLES:

see page 39

KEYBOARD INTERFACE

END READ KEYBOARD

JUMP TABLE ADDRESS: FC4B

ENTRY: none
EXIT: No Carry no character waiting.
Carry set we have an answer (could be an error).
Zero flag set No error.
A keyboard character if carry set.
error code if zero flag set.
DESCRIPTION:

This routine asks the keyboard what is happening. It returns a complex set of readings using the Carry and Zero
flags. It is important to check these in the correct order to trap all the conditions. First, check the carry flag to see if
the operation is complete. The next step is to check the zero flag for errors.

A combination of START and END read is the most effective way of reading the keyboard since it allows you to
check the keyboard for an interrupt command while some other process is ongoing (similar to catching CONTROL -
S when listing in SmartBasic).

EXAMPLES: Routine to check for abort within a process.
IGNORE:
CALL FCAS8 ;start read (only if not previously done).
JR NZ,ERROR
MORE:
CALL PRINTCHR ;send once character to the screen/printer.
CALL FC4B ;is there anything?
JR NC,MORE ;S0 let's print some more.
JR NZ,ERROR ;we have a keyboard error.
CP 19 ;is it CONTROL-S?
JR ZWAIT ;wait until other key pressed.
CP 3 ;is it CONTROL-C?
JR NZ,IGNORE ;if not ignore key but reset read.

;abort code goes here.
Routine to wait for a character and return it in B.

STARTGET:

CALL FCA8 ;enter here is need to start the keyboard.
JR NZ,ERROR ;start read.

WAITCHAR: ;enter here if start already done.

CALL FC4B ;end read.

JR NC ,WAITCHAR ;wait until acknowledge.

JR NZ, ERROR ;bad keyboard again.

LD B,A ;save character.

CALL FCAS8 ;restart read for next time.
JR NZ, ERROR

RET

FILE OPERATIONS

INITIALIZE FILE MANAGER

JUMP TABLE ADDRESS: FCBA
ENTRY: DE pointer to 3K of memory for FCB transfer buffers
HL pointer to 105 bytes for three File Control Blocks
EXIT: all registers preserved

DESCRIPTION:

This routine establishes three file control blocks and 3 1K buffers for the data sent/received through the FCB's. It
also sets up and sets to AVAILABLE three 35 byte file control blocks.

The EOS INIT sequence sets up three default FCB's and buffers. programmer, however, may wish to move these
elsewhere.

EXAMPLES:

This routine shows the default setup for the FCB's:

LD HL,D390 ;reserve 112 bytes for FCB's (round number?)
LD DE,D400 ;3K ending at EOOO.
CALL FCBA ;set up file manager.

FILE OPERATIONS

CHECK DIRECTORY FOR FILE

JUMP TABLE ADDRESS: FCFC
ENTRY: A device
HL pointer to file name

EXIT: Zero flag set means file found
A error code if non zero
BC DE  Start block of file if found
HL preserved

DESCRIPTION:

This routine ALWAYS uses the first file control block which is reserved for directory work. It sets the FCB to read
block 1 of the medium which is presumed to be the directory. It extracts the directory size from byte 12 of block 1 of
the media. The next step is to check if this is a directory by looking for the 55 AA 00 FF at bytes 13, 14, 15, and 16.
It then updates the file counters and begins searching for the file name. First, it looks for a BLOCKS LEFT entry; if
found, the routine aborts with error code 5 in A. Deleted entries are also skipped.

The next part is to compare the file name with the user supplied name in (HL). The search routine has the capability
of skipping the FILE TYPE byte based on a flag at FDD8. If non-zero, then the file type is expected to match. In
order to set this flag properly, you should use the FIND FILE 1 or FIND FILE 2 routines.

The routine continues scanning entries, reading additional blocks if necessary in order to find the file. If the file is
found it's start block is placed in registers BCDE and t